# NAG Library Routine Document

# D01BDF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

D01BDF calculates an approximation to the integral of a function over a finite interval $[a, b]$:

$$I = \int_a^b f(x)\,dx.$$

It is non-adaptive and as such is recommended for the integration of 'smooth' functions. These **exclude** integrands with singularities, derivative singularities or high peaks on $[a, b]$, or which oscillate too strongly on $[a, b]$.

## 2 Specification

```
SUBROUTINE D01BDF (F, A, B, EPSABS, EPSREL, RESULT, ABSERR)
REAL (KIND=nag_wp) F, A, B, EPSABS, EPSREL, RESULT, ABSERR
EXTERNAL           F
```

## 3 Description

D01BDF is based on the QUADPACK routine QNG (see Piessens *et al.* (1983)). It is a non-adaptive routine which uses as its basic rules, the Gauss 10-point and 21-point formulae. If the accuracy criterion is not met, formulae using 43 and 87 points are used successively, stopping whenever the accuracy criterion is satisfied.

This routine is designed for smooth integrands only.

## 4 References

Patterson T N L (1968) The Optimum addition of points to quadrature formulae *Math. Comput.* **22** 847–856

Piessens R, de Doncker–Kapenga E, Ûberhuber C and Kahaner D (1983) *QUADPACK, A Subroutine Package for Automatic Integration* Springer–Verlag

## 5 Arguments

1: F – REAL (KIND=nag_wp) FUNCTION, supplied by the user. *External Procedure*

F must return the value of the integrand $f$ at a given point.

The specification of F is:

```
FUNCTION F (X)
REAL (KIND=nag_wp) F
REAL (KIND=nag_wp) X
```

1: X – REAL (KIND=nag_wp) *Input*

*On entry*: the point at which the integrand $f$ must be evaluated.

F must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub) program from which D01BDF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

2: A – REAL (KIND=nag_wp) *Input*

*On entry*: $a$, the lower limit of integration.

3: B – REAL (KIND=nag_wp) *Input*

*On entry*: $b$, the upper limit of integration. It is not necessary that $a < b$.

4: EPSABS – REAL (KIND=nag_wp) *Input*

*On entry*: the absolute accuracy required. If EPSABS is negative, the absolute value is used. See Section 7.

5: EPSREL – REAL (KIND=nag_wp) *Input*

*On entry*: the relative accuracy required. If EPSREL is negative, the absolute value is used. See Section 7.

6: RESULT – REAL (KIND=nag_wp) *Output*

*On exit*: the approximation to the integral $I$.

7: ABSERR – REAL (KIND=nag_wp) *Output*

*On exit*: an estimate of the modulus of the absolute error, which should be an upper bound for $|I - \text{RESULT}|$.

## 6 Error Indicators and Warnings

There are no specific errors detected by D01BDF. However, if ABSERR is greater than

$$\max\{\text{EPSABS}, \text{EPSREL} \times |\text{RESULT}|\}$$

this indicates that the routine has probably failed to achieve the requested accuracy within 87 function evaluations.

## 7 Accuracy

D01BDF attempts to compute an approximation, RESULT, such that:

$$|I - \text{RESULT}| \leq tol,$$

where

$$tol = \max\{|\text{EPSABS}|, |\text{EPSREL}| \times |I|\},$$

and EPSABS and EPSREL are user-specified absolute and relative error tolerances. There can be no guarantee that this is achieved, and you are advised to subdivide the interval if you have any doubts about the accuracy obtained. Note that ABSERR contains an estimated bound on $|I - \text{RESULT}|$.

## 8 Parallelism and Performance

D01BDF is not threaded in any implementation.

## 9 Further Comments

The time taken by D01BDF depends on the integrand and the accuracy required.

# 10 Example

This example computes

$$\int_0^1 x^2 \sin(10\pi x)\, dx.$$

## 10.1 Program Text

```
!   D01BDF Example Program Text
!   Mark 26 Release. NAG Copyright 2016.

    Module d01bdfe_mod

!     D01BDF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Accessibility Statements ..
      Private
      Public                             :: f
!     .. Parameters ..
      Integer, Parameter, Public         :: nout = 6
    Contains
      Function f(x)

!       .. Use Statements ..
        Use nag_library, Only: x01aaf
!       .. Function Return Value ..
        Real (Kind=nag_wp)               :: f
!       .. Scalar Arguments ..
        Real (Kind=nag_wp), Intent (In) :: x
!       .. Intrinsic Procedures ..
        Intrinsic                        :: sin
!       .. Executable Statements ..
        f = (x**2)*sin(10.0E0_nag_wp*x01aaf(f)*x)

        Return

      End Function f
    End Module d01bdfe_mod
    Program d01bdfe

!     D01BDF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: d01bdf, nag_wp
      Use d01bdfe_mod, Only: f, nout
!     .. Implicit None Statement ..
      Implicit None
!     .. Local Scalars ..
      Real (Kind=nag_wp)                 :: a, abserr, b, epsabs, epsrel, result
!     .. Intrinsic Procedures ..
      Intrinsic                          :: abs, max
!     .. Executable Statements ..
      Write (nout,*) 'D01BDF Example Program Results'

      epsabs = 0.0E0_nag_wp
      epsrel = 1.0E-04_nag_wp
      a = 0.0E0_nag_wp
      b = 1.0E0_nag_wp

      Call d01bdf(f,a,b,epsabs,epsrel,result,abserr)

      Write (nout,*)
      Write (nout,99999) 'A      - lower limit of integration = ', a
```

```
      Write (nout,99999) 'B      - upper limit of integration = ', b
      Write (nout,99998) 'EPSABS - absolute accuracy requested = ', epsabs
      Write (nout,99998) 'EPSREL - relative accuracy requested = ', epsrel
      Write (nout,*)
      Write (nout,99997) 'RESULT - approximation to the integral = ', result
      Write (nout,99998) 'ABSERR - estimate to the absolute error = ', abserr
      Write (nout,*)

      If (abserr>max(epsabs,epsrel*abs(result))) Then
        Write (nout,*)                                                    &
          'Warning - requested accuracy may not have been achieved'
      End If

99999 Format (1X,A,F10.4)
99998 Format (1X,A,E9.2)
99997 Format (1X,A,F9.5)
    End Program d01bdfe
```

## 10.2 Program Data

None.

## 10.3 Program Results

```
 D01BDF Example Program Results

 A      - lower limit of integration =     0.0000
 B      - upper limit of integration =     1.0000
 EPSABS - absolute accuracy requested =  0.00E+00
 EPSREL - relative accuracy requested =  0.10E-03

 RESULT - approximation to the integral =  -0.03183
 ABSERR - estimate to the absolute error =  0.13E-10
```